

Kristallen tekenen met uw computer

door Jan Schilthuisen

Als u aan mineralogie doet en bovendien een home computer bezit, dan hebt u zich vast al eens afgevraagd, hoe het een voor het andere valt te gebruiken. Opslaan van de mineralenverzameling ligt het meest voor de hand. Het programmeren van een determinatiesysteem (bijv. aan de hand van de Gea-'rekenschijf') is al een stuk moeilijker. Een programma, dat kristalvormen kan berekenen en ze bovendien in mono of stereo kan tekenen, vergt meer dan de gemiddeld aanwezige wiskundige kennis.

Door Koen van de Moortel (Aartselaar, België) is in deze leemte voorzien. Van hem ontvingen wij zo'n programma, dat hij voor gebruik op een pocket-computer heeft ontworpen. Omdat het programma op een computer met meer geheugen wat gebruikersvriendelijker kan worden gemaakt, hebben wij dit aangevuld en vertaald voor de ZX Spectrum 48K. Omdat de BASIC van de Spectrum zo eenvoudig is, zal vertaling naar een andere BASIC niet veel problemen geven. De bijgevoegde illustraties laten slechts enkele van het onbegrensde aantal mogelijkheden zien. De kristallen worden door het programma als doorzichtige draadfiguren getekend. Om deze wat gemakkelijker herkenbaar te maken, zijn de van voren af zichtbare vlakken met de hand opgetekend.

Om met dit programma goed overweg te kunnen is het noodzakelijk iets af te weten van de gebruikelijke manieren om kristalvormen te beschrijven. Zie hiervoor bijv. E.A.J. Burke: De Miller-indices van kristalvlakken; Gea, vol. 18 (1985), nr. 3, pp. 107-112.

Programmastructuur

100-440: Invoer van de relatieve aslengten, de hoeken tussen de assen, de Miller-indices van de kristalvlakken en de gewenste grootte van de kristalvlakken.

450-740: Berekent de coördinaten van de hoekpunten van de verschillende kristalvlakken, alsof het een kubisch kristal betreft.

790-880: Minimaliseren van inmiddels overbodige array's. Assen verkorten of verlengen tot de gewenste relatieve lengte. Hoeken tussen de assen op de gewenste waarde instellen.

De coördinaten van de juiste eindvorm zijn nu beschikbaar. 900-1000: Dupliceren van de coördinaten, om deze te kunnen transformeren, zonder het origineel te verstoren. Invoer van de gewenste kijkhoek voor de te maken afbeelding.

1020-1140: Roteren van het geduplicateerde origineel in de gewenste stand voor de afbeelding.

1200-1215: Invoer van de gewenste schaal; mono- of stereotekening.


1220-1400: Plotten van de afbeelding op het beeldscherm; eventueel afdruk met printer.

2000-2010: Subroutine correctiefactor voor stereotekening.

8900-8950: Subroutine coördinaten origineel dupliceren.

9000-9290: Subroutine voor het transformeren van coördinaten t.b.v. de gewenste afbeelding.

9300-9340: Subroutine voor het omrekenen van graden naar radialen.



Afb. 1. Dit tekeningetje verschijnt op het scherm als hulp bij de invoer.

Toelichting op het programma


(Waar in de toelichting de ENTER-toets wordt bedoeld, is dit met aangegeven.)

Regels 100-440: geholpen door een tekening (afb. 1) wordt de gebruiker gevraagd naar de nodige gegevens voor het berekenen van de coördinaten. Ten eerste de relatieve aslengten. Als u een kubisch kristal gaat tekenen kunt u drie maal een '1' invoeren. Voor alle andere kristalsystemen is de onderlinge verhouding afhankelijk van het systeem en het betreffende mineraal. Die moet dus worden opgezocht. Dan de hoeken tussen de drie assen (alpha, beta en gamma). In het kubische, rhombische en tetragonale systeem zijn deze alle drie 90° . Voor monoklien en triklien moeten de hoeken worden opgezocht, omdat ze weer afhankelijk zijn van het mineraal.


Met het hexagonale en trigonale systeem is er iets bijzonders aan de hand. Het lijkt of we daarvoor een as tekort komen, omdat voor deze systemen altijd drie a-assen en een c-as worden gebruikt.

Voor het definiëren van de stand van de kristalvlakken is de a3-as echter overbodig. Wel moet de hoek (gamma)

Afb. 2. Stereogram van een kwartskristal. $\{10\bar{1}0\}$ wordt $\{100\}$, $d=1$. $\{01\bar{1}1\}$ wordt $\{011\}$, $d=1,5$. $\{10\bar{1}1\}$ wordt $\{01\}$, $d=1,8$. Gamma 120° .



tussen de assen a_1 en a_2 op 120° worden ingesteld. Omdat de hoeken worden gevraagd in graden, minuten en seconden, moet een hoek van bijv. 90° als volgt worden ingetoetst: $90^\circ 0' 0''$. Daarna worden de Miller-indices van de verschillende kristalvlakken opgevraagd (hier k , l en m genoemd). Deze indices kunnen bijvoorbeeld aan de afbeeldingen in handboeken worden ontleend. Ze kunnen niet op de gebruikelijke wijze worden geschreven, omdat de computer een getal als bijv. $\bar{1}$ niet kent. Het minteken komt vóór het betreffende cijfer. Dus bijv. $(\bar{1}10)$ wordt $-1\ 1\ 0$. Van de vier-cijferige indices voor hexagonale en trigonale kristallen (zie afb. 2 en 3) wordt consequent het **derde** cijfer verwaarloosd, omdat de a_3 -as niet aanwezig is. Bijv. $(20\bar{2}1)$ wordt dus: $2\ 0\ 0\ -1$.




Afb. 3a. Vesuvianiet (tetragonaal). $\{100\}$: $d=1$; $\{001\}$: $d=2$; $\{101\}$: $d=2,7$; $\{110\}$: $d=1,7$.

Afb. 3b. Calciet skalenöeder (trigonaal). $\{21\bar{3}1\}$ wordt $\{211\}$, $d=1$. Gamma 120° .


Naast de Miller-indices moet een waarde voor 'd' worden ingetoetst. 'd' is een maat voor de afstand tussen het kristalvlak en de oorsprong van het assenkruis. Het kiezen van een geschikte waarde voor 'd' is niet direct gemakkelijk en vereist enige ervaring. (De werkelijke afstand = $d/\sqrt{(k^2 + l^2 + m^2)}$.) Als 'd' te klein wordt gekozen, dan worden andere kristalvlakken verdrongen. Bij een te grote waarde zou het vlak buiten het kristal terecht komen (afb. 4). In beide gevallen geeft de computer een foutmelding en moet u opnieuw beginnen. Bij de illustraties zijn de Miller-indices en 'd' vermeld, zodat deze een indicatie geven van bruikbare waarden. Om tijd te sparen kan bij twijfel vaak eerst een eenvoudiger kristal worden opgegeven, om te testen of een vlak goed 'in beeld' blijft. Ook wordt tijd gespaard, door een kristal aan de onderkant af te sluiten met het basisvlak (001), (afb. 2).

In de regels 450-880 gaat de computer aan 't werk, zonder dat u hoeft in te grijpen. Doordat BASIC zo traag is, kan dat knap lang duren. Een kristal met 8 vlakken kost ca. 3 minuten, maar 20 vlakken vergen al een rekentijd van ruim een uur!

Als alles goed verloopt, zijn daarna de coördinaten van het te tekenen kristal beschikbaar en kunt u de gewenste kijkhoek bepalen door: 1e: roteren om de verticale as; 2e: voorover of achterover kantelen om een horizontale as, en tenslotte (als van de 2e mogelijkheid gebruik




Afb. 4. Een kubisch kristal, met een toenemende waarde van 'd' voor de $\{111\}$ -vlakken.



Afb. 5. Augiet (monoklien). A: zijaanzicht, B: vooraanzicht, C: bovenaanzicht, D: scheve projectie. $\{100\}$: $d=1$; $\{110\}$: $d=1,5$; $\{011\}$: $d=1$. Bèta: $74^\circ 10'$.

is gemaakt) nogmaals roteren om de verticale as. Door een juiste combinatie van deze drie mogelijkheden is het kristal van alle kanten te bekijken (afb. 5). De verrekening van deze gegevens vergt slechts weinig tijd. Het opragen van een nieuwe kijkhoek gaat dan ook vrij snel. Na het intoetsen van de gewenste schaal en mono- of



Afb. 6. Albiet (triklien), met de in te voeren data.
N.B. De afgebeelde kristallen zijn willekeurig georiënteerd en niet strikt volgens de gestandaardiseerde weergave in de handboeken.

stereotekening, wordt het kristal op het beeldscherm geplote en kan met de ZX-printer een kopie op papier worden gemaakt. Voor een stereografische afbeelding wordt het kristal twee maal getekend: eerst het rechter en daarna het linker beeld. Bovendien wordt dan onderaan een horizontale lijn afgedrukt, om de montage van de twee beelden te vergemakkelijken (afb. 2).

Verbeteringen

Geen enkel programma is direct volmaakt. In dit geval zullen pogingen tot verbetering vooral gericht zijn op verkorting van de rekentijd. Een andere computer kan wel wat sneller zijn, maar in BASIC blijft het toch sukkel. Als iemand kans ziet het programma in bijv. PASCAL te vertalen, dan zouden wij daar graag een listing van ontvangen.

*

"Shape" computerprogramma

Het in dit nummer gepubliceerde computerprogramma voor het tekenen van kristallen heeft het voordeel dat het op een betrekkelijk kleine computer kan draaien en dat het gratis is. Qua bedieningsgemak, snelheid en toepassingsmogelijkheden valt er echter nog wel iets te wensen. Een tweelingkristal valt er bijv. niet mee te berekenen. Voor meer professionele toepassing is in de Verenigde Staten het programma 'Shape' verkrijgbaar, dat op dit gebied alle mogelijkheden schijnt te bieden die men zich maar kan wensen. Er zijn twee versies: één voor de Apple II, met Apple Soft in Rom en 64K in Ram. De IIc kan alleen op het scherm tekenen. De andere versie draait op IBM p.c.'s en compatibles met 200K. De communicatie met hier verkrijgbare printers schijnt overigens niet altijd probleemloos te verlopen. 'Shape' kost 95 US-dollar en wordt geleverd door: Shape, 196 Beechwood Avenue Bogota, NJ 07603 New Jersey, U.S.A. Telefoon: 201-4873254.

J. Schilthuisen

*

Afb. 7. Listing van het programma voor het tekenen van kristallen voor de ZX Spectrum 48K (door K.v.d.Moortel en J.G.Schilthuisen).

```
50 REM ZX SPECTRUM 48 K
80 REM KRISTAL, © 1987 Koen v.
d.Moortel en Jan Schilthuisen
90 REM Versie GEA
```

```
100 REM INPUT KRISTAL
110 BORDER 5: INK 0: CLS
120 PRINT AT 2,2;"KRISTAL": PLO
T 120,159
130 DRAW 0,-88: DRAW -64,-64: P
LOT 120,71: DRAW 112,-56
140 INK 1: PRINT AT 20,1;"a1(a)
": INPUT "lengte a1? ";a1
150 PRINT AT 17,26;"a2(b)": INP
UT "lengte a2? ";a2
160 PRINT AT 2,16;"a3(c)": INPU
T "lengte a3? ";a3
170 INK 0
180 PLOT 150,55: DRAW -30,46,1:
PRINT INK 2;AT 10,19;"alfa"
190 INPUT "hoek alfa (grad,min,
sec) ";grad,min,sec: GO SUB 9340
: LET al=rad
200 PLOT 99,50: DRAW 21,51,-1:
PRINT INK 2;AT 11,6;"beta"
210 INPUT "hoek beta (grad,min,
sec) ";grad,min,sec: GO SUB 9340
: LET be=rad
220 PLOT 99,50: DRAW 51,5,1: PR
INT INK 2;AT 17,13;"gamma"
230 INPUT "hoek gamma in grad,m
in,sec. (120grad. voor hexagonaa
l) ";grad,min,sec: GO SUB 9340:
LET ga=rad
300 CLS: INPUT "aantal vlakken
? ";av: LET av=av-1
310 DIM k(av+1): DIM l(av+1): D
IM m(av+1): DIM d(av+1)
320 FOR i=0 TO av: PRINT "Ulak
";i+1;": ";
330 INPUT "k, l, m, ";k,l,m: LE
T k(i+1)=k: LET l(i+1)=l: LET m(i
+1)=m: PRINT k;l;m;
340 INPUT "D? ";d(i+1): IF d(i+
1)<0 THEN GO TO 340
350 PRINT INK 2;d(i+1)
360 NEXT i
370 INPUT "alles juist? ";a$: I
F a$="j" THEN GO TO 420
380 INPUT "fout in vlak nr? ";i
: INPUT "k, l, m, ";k,l,m: LET k
(i)=k: LET l(i)=l: LET m(i)=m: P
RINT INK 1;"Ulak ";i;": ";k;l;m;
390 INPUT "D? ";d(i): IF d(i)<0
THEN GO TO 390
400 PRINT INK 1;d(i)
410 GO TO 370
430 LET ret=av-1: LET rt=INT ((
ret*ret*ret/90)+1,5)
440 CLS: PRINT "Rekentijd ca "
;rt;" minuten"
```

```
450 REM SNIJPUNTEN
460 LET di=(av+1)*(av+1)
470 DIM p(30): DIM q(30): DIM r
(30): DIM u(di+1): DIM v(di+1):
DIM w(di+1): DIM t(av+1)
480 LET e=0: FOR h=0 TO av
490 LET f=0: FOR i=0 TO av: IF
i=h THEN GO TO 620
500 LET d1=l(h+1)*m(i+1)-l(i+1)
*m(h+1): LET d2=k(h+1)*m(i+1)-k(i
+1)*m(h+1): LET d3=k(h+1)*l(i+1)
-k(i+1)*l(h+1)
510 IF d1=0 AND d2=0 AND d3=0 T
HEN GO TO 620
520 LET g=0: FOR j=0 TO av: IF
j=h OR j=i THEN GO TO 610
530 LET d=k(j+1)*d1-l(j+1)*d2+m
(j+1)*d3: IF d=0 THEN GO TO 610
540 LET u=(d(h+1)*l(i+1)*m(j+1)
-l(j+1)*m(i+1))-d(i+1)*l(h+1)*
m(j+1)-l(j+1)*m(h+1)+d(j+1)*d1)
/d
550 LET v=(d(h+1)*k(j+1)*m(i+1)
-k(i+1)*m(j+1))+d(i+1)*k(h+1)*
m(j+1)-k(j+1)*m(h+1))-d(j+1)*d2)
/d
560 LET w=(d(h+1)*k(i+1)*l(j+1)
-k(j+1)*l(i+1))-d(i+1)*k(h+1)*
l(j+1)-k(j+1)*l(h+1)+d(j+1)*d3)
/d
570 FOR n=0 TO av: IF k(n+1)*u+
l(n+1)*v+m(n+1)*w>d(n+1)*1.00000
01 THEN GO TO 610
580 NEXT n: IF g=0 THEN GO TO 6
00
590 IF u=p(f) AND v=q(f) AND w=
r(f) THEN GO TO 610
```

```

600 LET p(f+1)=u: LET q(f+1)=v:
LET r(f+1)=w: LET f=f+1: LET g=
g+1: IF g=2 THEN GO TO 620
610 NEXT j: IF g=1 THEN LET f=f
-1
620 NEXT i
630 FOR i=1 TO f-3 STEP 2: LET
j=i
640 LET tol=.001: LET j=j+1: IF
ABS (p(j+1)-p(i+1))>tol OR ABS
(q(j+1)-q(i+1))>tol OR ABS (r(j+
1)-r(i+1))>tol THEN GO TO 640
650 IF INT (j/2)=j/2 THEN LET n
=j+1: GO TO 670
660 LET n=j-1
670 LET u=p(i+2): LET v=q(i+2):
LET w=r(i+2): LET p(i+2)=p(j+1)
: LET q(i+2)=q(j+1): LET r(i+2)=
r(j+1)
680 LET p(j+1)=u: LET q(j+1)=v:
LET r(j+1)=w: IF j=i+2 THEN GO
TO 710
690 LET u=p(i+3): LET v=q(i+3):
LET w=r(i+3): LET p(i+3)=p(n+1)
: LET q(i+3)=q(n+1): LET r(i+3)=
r(n+1)
700 LET p(n+1)=u: LET q(n+1)=v:
LET r(n+1)=w
710 NEXT i
720 FOR i=0 TO f/2-1: LET u(e+i
+1)=p(2*i+1): LET v(e+i+1)=q(2*i
+1): LET w(e+i+1)=r(2*i+1): NEXT
i
730 LET u(1+e+f/2)=p(f): LET v(
1+e+f/2)=q(f): LET w(1+e+f/2)=r(
f): LET t(h+1)=f/2+1: LET e=e+t(
h+1)
740 NEXT h

```

```

790 REM ORIGINEEL MODIFICEREN
800 DIM q(1): DIM r(1): LET av=
av+1: REM av nu werkelijke waard
e
810 LET totp=0: FOR h=1 TO av:
LET totp=totp+INT (t(h)+.5): NEX
T h
820 DIM p(totp,6): GO SUB 9000:
FOR h=1 TO totp
830 LET p(h,1)=v(h)*a2
840 LET p(h,2)=w(h)*a3
850 LET p(h,3)=u(h)*a1
860 GO SUB 9020: NEXT h
870 DIM u(1): DIM v(1): DIM w(1
)
880 BEEP 1,5

```

```

900 REM INPUT TEK.
910 DIM r(3): GO SUB 8900
920 CLS: LET vlag=0
930 PRINT "roteren om verticale
as (graden)""0=geen rotatie"
"-=" naar rechts""+=" naar links"
935 INPUT "Graden? ";grad: GO S
UB 9300
940 IF vlag=0 THEN LET r(1)=rad
: GO TO 960
950 LET r(3)=rad: GO TO 1000
960 CLS: PRINT "Kantelen om ho
rizontale as""0=niet kantelen"
"+=" achterover""-=" voorover"
970 INPUT "Graden? ";grad: IF g
rad=0 THEN GO TO 1000
980 GO SUB 9300: LET r(2)=rad
990 LET vlag=1: CLS: PRINT INK
2;"OPNIEUW "; GO TO 930
1000 CLS

```

```

1020 REM ROTATIES MET ORIGINEEL
1025 FOR h=1 TO totp
1030 IF r(1)=0 THEN GO TO 1060
1040 LET p(h,4)=(p(h,1)*COS (r(1
)))-(p(h,3)*SIN (r(1)))
1050 LET p(h,5)=(p(h,1)*SIN (r(1
)))+(p(h,3)*COS (r(1)))
1060 IF r(2)=0 THEN GO TO 1100
1070 LET z1=p(h,6)
1080 LET p(h,6)=(z1*COS (r(2)))-
(p(h,2)*SIN (r(2)))
1090 LET p(h,5)=(z1*SIN (r(2)))+
(p(h,2)*COS (r(2)))
1100 IF r(3)=0 THEN GO TO 1140
1110 LET x1=p(h,4)
1120 LET p(h,4)=(x1*COS (r(3)))-
(p(h,6)*SIN (r(3)))
1130 LET p(h,6)=(x1*SIN (r(3)))+
(p(h,5)*COS (r(3)))
1140 NEXT h: CLS

```

```

1200 REM TEKENEN
1210 INPUT "Schaal? ";s: INPUT "
mono of stereo? m/s ";s$
1215 LET corr=0: CLS
1220 LET e=0: FOR h=1 TO av: FOR
i=1 TO t(h): LET e=e+1
1230 LET xe=INT ((p(e,4)+(corr*
p(e,6)))*s)+.5)
1240 LET ye=INT ((p(e,5)*s)+.5)
1250 IF xe<(-127) OR xe>127 OR y
e<(-87) OR ye>87 THEN CLS: PRIN
T "Schaal te groot": GO TO 1210
1290 LET xe=xe+127: LET ye=ye+87
1300 IF i=1 THEN PLOT xe,ye: LET
xb=xe: LET yb=ye: GO TO 1320
1310 DRAW xe-xb,ye-yb: LET xb=xe
: LET yb=ye
1320 NEXT i: NEXT h
1330 IF s$="s" THEN PLOT 0,0: DR
AW 255,0
1340 INPUT "Copy? ";c$: IF c$="j
" THEN COPY
1350 IF s$="s" THEN LET s$="m":
CLS: GO TO 2000
1360 INPUT "Nog een tekening? ";
a$: IF a$="j" THEN GO TO 900
1400 STOP

```

```

2000 REM STEREO
2010 LET corr=.05: PLOT 0,0: DRA
W 255,0: GO TO 1220

```

```

8900 REM HERSTEL MODEL
8910 FOR h=1 TO totp
8920 LET p(h,4)=p(h,1)
8930 LET p(h,5)=p(h,2)
8940 LET p(h,6)=p(h,3)
8950 NEXT h: RETURN

```

```

9000 REM ROTATIES ASSEN
9010 DIM r(3): LET r(1)=a1-PI/2:
LET r(2)=b1-PI/2: LET r(3)=ga-P
I/2
9015 RETURN
9020 IF r(1)=0 THEN GO TO 9100
9030 LET p(h,2)=(-p(h,1))*SIN r(
1)+p(h,2)
9040 LET p(h,1)=p(h,1)*COS r(1)
9100 IF r(2)=0 THEN GO TO 9200
9120 LET p(h,2)=(-p(h,3))*SIN r(
2)+p(h,2)
9130 LET p(h,3)=p(h,3)*COS r(2)
9200 IF r(3)=0 THEN GO TO 9290
9220 LET p(h,1)=(-p(h,3))*SIN r(
3)+p(h,1)
9230 LET p(h,3)=p(h,3)*COS r(3)
9290 RETURN

```

```

9300 REM GRADEN NAAR RAD
9310 LET min=0: LET sec=0
9340 LET rad=(grad+min/60+sec/36
00)/180*PI: RETURN
9998 STOP

```

```

9999 SAVE "kristal" LINE 1: SAVE
"kristal" LINE 1: VERIFY "krist
al": VERIFY "kristal"

```